Biophysical Journal, Volume *117*

Supplemental Information

Taking Simultaneous Snapshots of Intrinsically Disordered Proteins in Action

Marco Schiavina, Maria Grazia Murrali, Letizia Pontoriero, Valerio Sainati, Rainer Kümmerle, Wolfgang Bermel, Roberta Pierattelli, and Isabella C. Felli



Figure S1. CON//HN pulse sequence scheme.

Narrow and wide black rectangles represent $\pi/2$ and π non-selective pulses; narrow and wide black symbols represent $\pi/2$ and π band-selective pulses (Q5 and Q3 shapes of 300 and 231 µs respectively), hatched pulses are adiabatic inversion pulse on C′ and C^{α} (smoothed Chirp 500 µs 80 kHz sweep and 20% smooting).

The following phase cycling was employed for the CON//HN: ϕ_1 = x, -x ; ϕ_2 = 2(x), 2(-x); ϕ_3 = 4(x), 4(-x), ϕ^{IP} = x ; ϕ^{AP} = -y and $\phi_{rec}{}^{CON}$= x, -x, x, -x, -x, x, -x, x for the 2D-CON and ϕ_4 = x, –x, ϕ_5 = 2(x), 2(-x) ϕ_6 = 4(x) 4(-x) and $\phi_{rec}{}^{HN}$ = x, -x, x, -x, -x, x, -x, x for the 2D-HN. Quadrature detection in the indirect dimension was achieved through the STATES-TPPI approach incrementing phase ϕ_1 (CON) and ϕ_4 (HN).

The length of the delays was: d_1= 1.7 s; Δ= 16.6 ms; Δ_1 = 4.5 ms; Δ_2= 2.7 ms; ε = ½ duration of adiabatic pulse (250 µs); ε´= t_1(0) + duration of adiabatic pulse (500 µs).

Virtual decoupling of the C′-C^{α} coupling in the 2D-CON was achieved by acquiring for each increment both the IP and AP component of the signals and combining them.



Figure S2. CON//btNH pulse sequence scheme.

Narrow and wide black rectangles represent $\pi/2$ and π non-selective pulses; narrow and wide black symbols represent $\pi/2$ and π band-selective pulses (Q5 and Q3 shapes of 300 and 231 µs respectively), hatched pulses are adiabatic inversion pulse on C′ and C^{α} (smoothed Chirp 500 µs 80 kHz sweep and 20% smooting), the shaded shapes are Bip pulses.

The following phase cycling was employed for the CON//btNH: ϕ_1 = x, -x; ϕ_2 = 2(x), 2(-x); ϕ_3 = 4(x), 4(-x) and ϕ_{rec}^{CON}= x, -x, x, -x, -x, x, -x, x for the 2D-CON and ϕ_4 =y, -y, x, -x; ϕ_5 = y ϕ_6 = x and ϕ_{rec}^{btNH} = x, -x, -y, y for the 2D-btNH. Quadrature detection in the indirect dimension was achieved through the STATES-TPPI approach incrementing phase ϕ_1 for the CON and through the Echo/Antiecho approach by incrementing phase ϕ_4 and ϕ_5.

The length of the delays was: d_1= 2.35 s; Δ= 16.6 ms; Δ_1 = 4.5 ms; Δ_2= 2.7 ms; ε= $t_1(0)$ + duration of adiabatic pulse (500 µs).

Virtual decoupling of the C'-C^{α} coupling in the 2D-CON was achieved by acquiring for each increment both the IP and AP component of the signals. The same approach was also employed to achieve heteronuclear decoupling (C'-N) during acquisition of the ^{15}N FIDs in order to preserve the C' polarization for the subsequent CON experiment. Band selective C^{α} decoupling was achieved using a Q3 pulse in a P5M4 supercycle.



Figure S3. CON//(H)CAN pulse sequence scheme.

Narrow and wide black rectangles represent π/2 and π non-selective pulses; narrow and wide black symbols represent π/2 and π band-selective pulses (Q5 and Q3 shapes of 300 and 231 µs respectively), hatched pulses are adiabatic inversion pulse on C′ and C^{α} (smoothed Chirp 500 µs 80 kHz sweep and 20% smooting).

The following phase cycling was employed for the CON//(H)CAN: ϕ_1 = x, -x; ϕ_2 = 2(x), 2(-x); ϕ_3 = 4(x), 4(-x) ; ϕ^{IP} = x ; ϕ^{AP} = -y and ϕ_{rec}^{CON}= x, -x, x, -x, -x, x, -x, x for the 2D-CON and ϕ_4 = x, –x, ϕ_5 = 4(x), 4(-x) ϕ_6 = 8(x) 8(-x) ; ϕ'^{IP} = 2(x), 2(-x) ; ϕ'^{AP} = 2(y), 2(-y) and $\phi_{rec}^{(H)CAN}$ = x, -x, -x, x, -x, x, x, -x for the 2D-(H)CAN. Quadrature detection in the indirect dimension was achieved through STATES-TPPI approach incrementing phase ϕ_1 (CON) and ϕ_4 ((H)CAN).

The length of the delays was: d_1= 2.5 s; Δ= 12.4 ms; Δ_1 = 4.5 ms; Δ_2= 1.7 ms; Δ_3= 1.9 ms; Δ_4= 1.4 ms; Δ_5= 4.2 ms; Δ_6= 16.6 ms. t_a, t_b and t_c were used to achieve the constant time mode for the $^{13}C^{\alpha}$ indirect dimension.

Virtual decoupling of the C′-C^{α} coupling in the CON was achieved by acquiring for each increment both the IP and AP component of the signals. The same approach was also employed to achieve heteronuclear decoupling (C'-N) during acquisition of the ^{15}N FIDs in order to preserve the C' polarization for the subsequent CON experiment. Band selective C^{α} decoupling was achieved using a Q3 pulse in a P5M4 supercycle.

Pulse sequence CON//HN

```
;ut_con_fhsqchn
;avance-version

;Dataset 1 (F1)
;fhsqcf3gpph
;avance-version (12/01/11)
;2D H-1/X correlation via double inept transfer
;
;      F1(HN) -> F3(N,t1) -> F1(HN,t2)
;
;phase sensitive
;with decoupling during acquisition
;
;(S. Mori, C. Abeygunawardana, M. O'Neil-Johnson & P.C.M. van Zijl,
;   J. Magn. Reson. B 108, 94-98 (1995) )
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


;Dataset 2 (F2)
;CON
;2D sequence with
;   13C detected correlation for triple resonance using
;      inept transfer steps
;
;      F1(C=O) -> F3(N,t1) -> F1(C=O,t2)
;
;on/off resonance 13C pulses using shaped pulses
;phase sensitive (t1)
;using IPAP scheme for virtual decoupling
;(use parameterset C_CON_IASQ)
;
;W. Bermel, I. Bertini,L. Duma, I.C. Felli, L. Emsley,  R. Pierattelli,
;   P.R. Vasos, Angew. Chem. Int. Ed. 44, 3089-3092 (2005)
;(L. Duma, S. Hediger, A. Lesage & L. Emsley,
;   J. Magn. Reson. 164, 187-195 (2003) )
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


prosol relations=<triple>


#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"p2=p1*2"
"p22=p21*2"
"d11=30m"
"d12=20u"
"d26=1s/(cnst4*4)"

"d22=4.5m"
"d23=16.6m"

"in31=inf1/2"
"in32=in2/2"
```

```
"d31=3u"
"d32=3u"


"DELTA=d19-p22/2"
"DELTA1=d26-p16-d16-p27*3-d19*5-p1*2/PI"
"DELTA2=d26-p16-d16-p27*2-p0-d19*5-de-8u"
"DELTA3=d31+larger(p2,p8)/2"
"DELTA4=p21*2/PI"
"DELTA5=d32*2+p8"
"DELTA6=d23/2-p14/2"
"DELTA7=d23-d22-p14"

"TAU=d26-p16-d16-4u"


"l0=1"

"l3=td1/2"


"spoff2=0"
"spoff3=0"
"spoff5=bf2*((cnst22-cnst21)/1000000)"
"spoff8=0"
"spoff13=bf2*((cnst26-cnst21)/1000000)"


1 4u ze1
  4u ze2

2 d11 do:f1 do:f3
  3m
3 12m
4 6m
5 d1

; fhsqc

  20u fq=cnst47(bf ppm):f3
  d12 pl1:f1 pl3:f3
  50u UNBLKGRAD

  (p1 ph11)
  4u
  p16:gp1
  d16
  TAU pl3:f3
  (center (p2 ph11) (p22 ph16):f3 )
  4u
  TAU
  p16:gp1
  d16
  (p1 ph12)

  4u
  p16:gp2
  d16
  (p8:sp13 ph17):f2
  4u

  (p21 ph13):f3
  DELTA3
  (p22 ph13):f3
  DELTA4
  d31

  (center (p2 ph15) (p8:sp13 ph11):f2 )
```

```
  d31
  DELTA4
  (p22 ph14):f3
  DELTA3
  (p21 ph14):f3

  4u
  p16:gp2
  d16

  (p1 ph17)
  p16:gp3
  d16 pl18:f1
  DELTA1
  p27*0.231 ph18
  d19*2
  p27*0.692 ph18
  d19*2
  p27*1.462 ph18
  DELTA
  (p22 ph11):f3
  DELTA
  p27*1.462 ph19
  d19*2
  p27*0.692 ph19
  d19*2
  p0*0.231 ph19
  4u
  p16:gp3
  d16
  4u BLKGRAD
  DELTA2 pl16:f3

  goscnp1 ph31 cpd3:f3


; con

  4u do:f3
  50u UNBLKGRAD

  p16:gp5
  d16

  20u pl19:f1 pl3:f3
  20u fq=cnst21(bf ppm):f2
  20u fq=cnst57(bf ppm):f3
  d12 cpds1:f1

  (p13:sp2 ph1):f2
  d23
  (center (p14:sp3 ph1):f2 (p22 ph1):f3 )
  d23
  (p13:sp8 ph2):f2

  p16:gp4
  d16

  (p21 ph3):f3
  d32
  (p8:sp13 ph5):f2
  d32
  (p22 ph1):f3
  DELTA5
  (p21 ph4):f3

  if "l0 %2 == 1"
     {
     (p13:sp2 ph1):f2
     DELTA6
```

```
      (p14:sp5 ph1):f2
      DELTA6
      (center (p14:sp3 ph1):f2 (p22 ph1):f3 )
      DELTA6
      (p14:sp5 ph1):f2
      DELTA6 pl16:f3
      }
   else
      {
      (p13:sp2 ph6):f2
      d22
      (p14:sp5 ph1):f2
      DELTA7
      (center (p14:sp3 ph1):f2 (p22 ph1):f3 )
      DELTA6
      DELTA6 pl16:f3
      (p14:sp5 ph1):f2
      }

   4u BLKGRAD
   go2=2 ph30 cpd3:f3

   d11 do:f1 do:f3 wr2 #1 if2 #1 zd2

;exp_f2 ipap
   3m iu0
   lo to 3 times 2

   3m wr1 #0 if1 #0 zd1

;exp_f1 phase
   3m ip13
   3m ip16
;exp_f2 phase
   3m ip3
   lo to 4 times 2

;exp_f1 delay
   3m id31
;exp_f2 delay
   3m id32
   lo to 5 times l3

exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 2 2
ph6=3

ph11=0
ph12=1
ph13=0 2
ph14=0 0 0 0 2 2 2 2
ph15=0 0 2 2
ph16=0
ph17=2
ph18=1
ph19=3

ph30=0 2 0 2 2 0 2 0
ph31=0 2 0 2 2 0 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl16: f3 channel - power level for CPD/BB decoupling
;pl18: f1 channel - power level for 3-9-19-pulse (watergate)
```

```
;pl19: f1 channel - power level for CPD/BB decoupling
;sp2 : f1 channel - shaped pulse  90 degree  (on resonance)
;sp3 : f1 channel - shaped pulse 180 degree  (on resonance)
;sp5 : f1 channel - shaped pulse 180 degree  (Ca off resonance)
;sp8 : f1 channel - shaped pulse  90 degree  (on resonance)
;                  for time reversed pulse
;sp13: f1 channel - shaped pulse 180 degree  (adiabatic)
;p0 : f1 channel -  90 degree pulse at pl18
;                   use for fine adjustment
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p8 : f1 channel - 180 degree shaped pulse for inversion (adiabatic)
;p13: f1 channel -  90 degree shaped pulse
;p14: f1 channel - 180 degree shaped pulse
;p16: homospoil/gradient pulse                          [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p27: f1 channel -  90 degree pulse at pl18
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                                [30 msec]
;d12: delay for power switching                         [20 usec]
;d16: delay for homospoil/gradient recovery
;d19: delay for binomial water suppression
;d22: 1/(4J(COCa))                                      [4.5 msec]
;d23: 1/(4J(NCO))                                       [16.6 msec]
;d26: 1/(4J(NH))
;d31: incremented delay (2D, exp 1)                     [3 usec]
;d32: incremented delay (2D, exp 2)                     [3 usec]
;cnst4: = J(NH)
;cnst21: CO chemical shift (offset, in ppm)
;cnst22: Calpha chemical shift (offset, in ppm)
;cnst26: Call chemical shift (offset, in ppm)           [101 ppm]
;cnst47: N(H) chemical shift (offset, in ppm)
;cnst57: N chemical shift (offset, in ppm)
;o1p: CO chemical shift (cnst21)
;l0: flag to switch between inphase and antiphase
;l3 : loop for 2D experiment = td1/2
;inf1: 1/SW(N) = 2 * DW(N)
;in2 : 1/SW(N) = 2 * DW(N) (inf1 of exp 2, N all)
;in31: 1/(2 * SW(N)) =  DW(N)
;nd31: 2
;in32: 1/(2 * SW(N)) =  DW(N)
;nd32: 2
;ns: 8 * n
;ds: >= 32
;td1: number of experiments in F1
;       (number of experiments after IPAP processing)
;FnMODE: States-TPPI (or TPPI) in F1
;cpds1: decoupling according to sequence defined by cpdprg1
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd1: f1 channel - 90 degree pulse for decoupling sequence
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz1: 50%
;gpz2: 80%
;gpz3: 31%
;gpz4: 19%
;gpz5: 60%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100

;use AU-program splitcomb [ipap 2] to process data

;$Id: $
```

Pulse sequence CON//btNH

```
;ut_con_n_btnh
;avance-version

;Dataset 1 (F1)
;c_con_iasq
;avance-version
;CON
;2D sequence with
;   13C detected correlation for triple resonance using
;      inept transfer steps
;
;      F1(C=O) -> F3(N,t1) -> F1(C=O,t2)
;
;on/off resonance 13C pulses using shaped pulses
;phase sensitive (t1)
;using IPAP scheme for virtual decoupling
;(use parameterset C_CON_IASQ)
;
;W. Bermel, I. Bertini,L. Duma, I.C. Felli, L. Emsley,  R. Pierattelli,
;   P.R. Vasos, Angew. Chem. Int. Ed. 44, 3089-3092 (2005)
;(L. Duma, S. Hediger, A. Lesage & L. Emsley,
;   J. Magn. Reson. 164, 187-195 (2003) )
;
;
;Dataset 2 (F2)
;dummy dataset
;
;
;Dataset 3 (F3)
;nb_hntrosy_f2ig
;avance-version (17/02/10)
;2D 15N shift correlation
;
;      F2(HN,t1) -> F3(N,t2)
;
;with refocussing of chemical shifts
;phase sensitive
;using semi-constant time in t1
;(use parameterset )
;
;(R. Schnieders, C. Richter, S. Warhaut, V. de Jesus, S. Keyhani,
;   E. Duchardt-Ferner, H. Keller, J. WÃ¶hnert, L.T. Kuhn, A.L. Breeze,
;   W. Bermel, H. Schwalbe & B. FÃ¼rtig, ;   J. Biomol. NMR 69, 31â€“44 (2017) )
;(K. Takeuchi, H. Arthanari, I. Shimada & G. Wagner, J. Biomol. NMR 63, 1â€“9 (2015) )
;(K. Takeuchi, H. Arthanari, M. Imai & G. Wagner, J. Biomol. NMR 64, 143â€“151 (2016) )
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


prosol relations=<triple_c>


#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"p22=p21*2"
"d11=30m"
"d12=20u"
"d26=1s/(cnst4*4)"

"d22=4.5m"
```

```
"d23=16.6m"


"p42=(bwfac42/(cnst55*cnst52*bf2))*1000000"
"spw42=plw2/((p42*90.0)/(p3*totrot42))*((p42*90.0)/(p3*totrot42))*(integfac42*integ-
fac42)"
"spoal42=0.5"
"spoff42=bf2*(cnst54/1000000)-o2"

"p43=(bwfac43/(cnst55*cnst53*bf2))*1000000"
"spw43=plw2/((p43*90.0)/(p3*totrot43))*((p43*90.0)/(p3*totrot43))*(integfac43*integ-
fac43)"
"spw44=plw2/((p43*90.0)/(p3*totrot44))*((p43*90.0)/(p3*totrot44))*(integfac44*integ-
fac44)"
"spoal43=0"
"spoal44=1"
"spoff43=bf2*(cnst54/1000000)-o2"
"spoff44=bf2*(cnst54/1000000)-o2"


"in31=inf1/2"
"in32=in2/2"

"d31=3u"
"d32=in32/2-p39/2"


"d60=d1-(d61+d59)*4-d58"


"DELTA=d31"
"DELTA1=d23/2-p12/2"
"DELTA2=d23-d22-p12"
"DELTA11=d26-larger(p22,p42)/2-p16-d16"
"DELTA12=d26-larger(p22,p42)/2-p16-d16-p21*2/PI"
"DELTA13=d26-larger(p22,p42)/2-p16-d16-de"
"DELTA14=d23/2-p12/2"


"l0=0"
"l3=l22/2"
"l10=1"
"l13=td1/(8*l22)"
"l30=0"


"spoff4=bf1*((cnst26-cnst22)/1000000)"
"spoff13=bf1*((cnst26-cnst21)/1000000)"
"spoff23=0"
"spoff24=0"
"spoff25=0"
"spoff26=bf1*((cnst22-cnst21)/1000000)"
"spoff27=bf1*((cnst21-cnst22)/1000000)"


1 4u ze1
  4u ze3
2 d11 do:f3
  3m
3 3m
4 3m
5 6m
6 9m
7 12.02m
8 3m

  d60
  50u UNBLKGRAD

goscnp2 ph31
```

```
  p16:gp2
  d16

;n_trhn

  d12 pl19:f1 pl3:f3
  20u fq=cnst22(bf ppm):f1

  "cnst31=(l30%2)*90"

  3m ip18+cnst31

11 d61

  (p39:sp4 ph17):f1

  if "l0 %2 == 1"
     {
     (p43:sp44 ph13+ph18):f2
     }
  else
     {
     (p43:sp44 ph14+ph18):f2
     }
  d32
  (p39:sp4 ph1+ph18):f1
  d32

  (p21 ph15+ph18):f3
  DELTA11
  p16:gp3
  d16
  4u
  (center (p42:sp42 ph1+ph18):f2 (p22 ph1+ph18):f3 )
  4u
  p16:gp3
  d16
  DELTA11
  (p21 ph1+ph18):f3

  (p43:sp43 ph2+ph18):f2
  DELTA12
  p16:gp4
  d16
  4u
  (center (p42:sp42 ph1+ph18):f2 (p22 ph1+ph18):f3 )
  4u
  p16:gp4
  d16
  DELTA13
  4u BLKGRAD
  (p43:sp44 ph16+ph18):f2

  if "l30 %2 == 0"
     {
     DELTA14
     (p12:sp27 ph1):f1
     DELTA14
     (p22 ph1):f3
     DELTA14
     (p12:sp27 ph17):f1
     DELTA14 pl19:f1
     }
  else
     {
     (p12:sp27 ph1):f1
     DELTA14
     DELTA14
     (p22 ph1):f3
```

```
      (p12:sp27 ph17):f1
      DELTA14
      DELTA14 pl19:f1
      }

  goscnp3 ph30 cpd1:f1

  4u do:f1
  lo to 11 times 4

  50u UNBLKGRAD

  p16:gp5
  d16

;con

  20u pl3:f3
  20u fq=cnst21(bf ppm):f1

  (p11:sp23 ph1)
  d23
  (center (p12:sp24 ph1) (p22 ph1):f3 )
  d23
  (p11:sp25 ph2)

  p16:gp1
  d16

  (p21 ph3):f3
  d31
  (center (p8:sp13 ph5):f1 (p44:sp3 ph1):f2 )
  d31
  (p22 ph1):f3
  DELTA
  (center (p8:sp13 ph1):f1 (p44:sp3 ph7):f2 )
  DELTA
  (p21 ph4):f3

  if "l10 %2 == 1"
     {
     (p11:sp23 ph1)
     DELTA1
     (p12:sp26 ph1)
     DELTA1
     (center (p12:sp24 ph1) (p22 ph1):f3 )
     DELTA1
     (p12:sp26 ph1)
     DELTA1 pl16:f3
     }
  else
     {
     (p11:sp23 ph6)
     d22
     (p12:sp26 ph1)
     DELTA2
     (center (p12:sp24 ph1) (p22 ph1):f3 )
     DELTA1
     DELTA1 pl16:f3
     (p12:sp26 ph1)
     }

  4u BLKGRAD

  go1=2 ph31 cpd3:f3

  d11 do:f3 wr1 #0 if1 #0 zd1
  3m wr2 #1 if2 #1 zd2

;exp_f1 ipap
```

```
  3m iu10
  lo to 3 times 2

;exp_f1 phase
  3m ip3
  lo to 4 times 2

;exp_f1 delay
  3m id31
  lo to 5 times l13

  3m wr3 #2 if3 #2 zd3

  3m iu30
  lo to 6 times 2

;exp_f2 phase
  3m ip15*2
  3m ip16*2
  3m iu0
  lo to 7 times 2

;exp_f2 delay
  3m id32
  3m
  20u
  3m
  3m
  lo to 8 times l3


exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 2 2
ph6=3
ph7=2
ph13=1 3 0 2
ph14=3 1 0 2
ph15=1
ph16=0
ph17=2
ph18=0
ph30=0 2 3 1
ph31=0 2 0 2 2 0 2 0


;pl1 : f1 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl16: f3 channel - power level for CPD/BB decoupling
;pl19: f1 channel - power level for CPD/BB decoupling
;sp3 : f2 channel - shaped pulse 180 degree (Bip720,50,20.1)
;sp4 : f1 channel - shaped pulse 180 degree (Bip720,100,10.1)
;sp13: f1 channel - shaped pulse 180 degree  (adiabatic)
;sp23: f1 channel - shaped pulse  90 degree  (on resonance)
;sp24: f1 channel - shaped pulse 180 degree  (on resonance)
;sp25: f1 channel - shaped pulse  90 degree  (on resonance)
;                   for time reversed pulse
;sp26: f1 channel - shaped pulse 180 degree  (Ca off resonance)
;sp27: f1 channel - shaped pulse 180 degree  (C=O off resonance)
;sp42: f2 channel - shaped pulse 180 degree (Reburp.1000)
;sp43: f2 channel - shaped pulse  90 degree (Eburp2.1000)
;sp44: f2 channel - shaped pulse  90 degree (Eburp2tr.1000)
;p8 : f1 channel - 180 degree shaped pulse for inversion (adiabatic)
;p11: f1 channel -  90 degree shaped pulse
;p12: f1 channel - 180 degree shaped pulse
```

```
;p16: homospoil/gradient pulse                          [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p39: f1 channel - 180 degree shaped pulse for refocussing
;                     Bip720,100,10.1 (160us at 600.13 MHz)
;p42: f2 channel - 180 degree shaped pulse for refocussing
;                     Reburp.1000               (1.4ms at 600.13 MHz)
;p43: f2 channel -  90 degree shaped pulse for excitation
;                     Eburp2.1000/Eburp2tr.1000 (1.7ms at 600.13 MHz)
;p44: f2 channel - 180 degree shaped pulse for refocussing
;                     Bip720,50,20.1            (200us at 600.13 MHz)
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                                [30 msec]
;d12: delay for power switching                         [20 usec]
;d16: delay for homospoil/gradient recovery
;d22: 1/(4J(COCa))                                      [4.5 msec]
;d23: 1/(4J(NCO))                                       [16.6 msec]
;d26: 1/(4J)NH
;d31: incremented delay (2D, exp 1))                    [3 usec]
;d32: incremented delay (2D, exp 3)): in32/2-p39/2
;d58: = AQ of exp 2 (1H)
;d59: = AQ of exp 3 (15N)
;d60: relaxation delay as executed
;d61: relaxation delay of exp 3
;cnst4: = J(NH)
;cnst21: CO chemical shift (offset, in ppm)
;cnst22: Calpha chemical shift (offset, in ppm)
;cnst26: Call chemical shift (offset, in ppm)          [101 ppm]
;cnst43: compensation of chemical shift evolution during p43
;           Eburp2.1000: 0.69
;cnst52: scaling factor for p42 to compensate for transition region
;           Reburp.1000: 1.426
;cnst53: scaling factor for p43 to compensate for transition region
;           Eburp2.1000: 1.000
;cnst54: H(N) chemical shift (offset, in ppm)
;cnst55: H(N) bandwidth (in ppm)
;o1p: CO chemical shift (cnst21)
;l0 : flag to switch between odd and even increments (exp 3)
;l3 : loop for 2D exp 1 = l22/2
;l13: loop for 2D exp 3 = td1/(8*l22)
;l10: flag to switch between inphase and antiphase (exp 1)
;l22: number of experiments in F1 for exp 3 (after IPAP processing)
;l30: flag to switch between inphase and antiphase (exp 3)
;inf1: 1/SW(N) = 2 * DW(N)
;in2 : 1/SW(H) = 2 * DW(H) (inf1 of exp 3)
;in31: 1/(2 * SW(N)) =  DW(N)
;nd31: 2
;in32: 1/(2 * SW(H)) = DW(H)
;nd32: 2
;ns: 8 * n
;ds: >= 32
;td1: number of experiments in F1
;       (number of experiments after IPAP processing)
;FnMODE: States-TPPI (or TPPI) in F1
;cpd1: decoupling according to sequence defined by cpdprg1: p5m4sp180
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd1: f1 channel - 180 degree pulse for decoupling sequence
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence
;cpdprg1: p5m4sp180


;for z-only gradients:
;gpz1: 50%
;gpz2: 70%
;gpz3: 31%
;gpz4: 19%
;gpz5: 45%

;use gradient files:
;gpnam1: SMSQ10.100
```

```
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100


;use AU-program splitcomb [ipap 2] to process data

;Processing (exp 3)

;PHC0(F1): 90
;PHC1(F1): -180
;FCOR(F1): 1


;$Id: $
```

Pulse sequence CON//(H)CAN

```
;ut_con_n_(h)can
;avance-version

;Dataset 1 (F1)
;n_hcan.2
;avance-version
;2D sequence with
;   15N detected correlation for triple resonance using
;      inept transfer steps
;
;      F3(Ha) -> F2(Ca,t1) -> F1(N,t2)
;
;on/off resonance 13C pulses using shaped pulses
;phase sensitive (t1)
;using constant time in t1
;(use parameterset )
;
;M. Gal, K. A. Edmonds, A. G. Milbradt, K. Takeuchi & G. Wagner,
;   J Biomol NMR 51, 497-504 (2011)
;(K. Takeuchi, G. Heffron, Z.-Y. J. Sun, D. P. Frueh & G. Wagner,
;   J Biomol NMR 47, 271-282 (2010))
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


;Dataset 2 (F2)
;c_con_iasq
;avance-version
;CON
;2D sequence with
;   13C detected correlation for triple resonance using
;      inept transfer steps
;
;      F2(C=O) -> F1(N,t1) -> F2(C=O,t2)
;
;on/off resonance 13C pulses using shaped pulses
;phase sensitive (t1)
;using IPAP scheme for virtual decoupling
;(use parameterset )
;
;W. Bermel, I. Bertini,L. Duma, I.C. Felli, L. Emsley,  R. Pierattelli,
;   P.R. Vasos, Angew. Chem. Int. Ed. 44, 3089-3092 (2005)
;(L. Duma, S. Hediger, A. Lesage & L. Emsley,
;   J. Magn. Reson. 164, 187-195 (2003) )
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


prosol relations=<triple>


#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"p2=p1*2"
"p22=p21*2"
"d3=1s/(cnst2*cnst12)"
```

```
"d4=1s/(cnst2*4)"
"d11=30m"
"d12=20u"

"d22=4.5m"
"d23=12.4m"
"d25=16.6m"
"d27=14.3m"


"d32=3u"
"d31=d27/2-d3-p22/2"
"d41=d27/2-larger(p14,p22)/2"
"d51=d23-d27/2-p14/2-p2/2"


"in31=inf1/4"
"in32=in2/2"

"in41=in32"
"in51=in32"

"td1=tdmax(td1,d41*2,in42)"


"d1=(d60+d58)*3 +d59+50m"


"DELTA1=d27-d23-p2/2"
"DELTA2=d3-p14/2-p22/2"
"DELTA3=d25/2"
"DELTA4=d23-d25/2-p14"
"DELTA5=d25-d23"
"DELTA6=d25/2-d12+p1*2/PI-de-4u"
"DELTA7=d32*2+p8"
"DELTA8=d23/2-p14/2"
"DELTA9=d23-d22-p14"


"spoff2=0"
"spoff3=0"
"spoff5=bf2*((cnst21-cnst22)/1000000)"
"spoff7=bf2*((cnst22-cnst21)/1000000)"
"spoff8=0"
"spoff13=bf2*((cnst26-cnst21)/1000000)"
"spoff23=bf2*((cnst23-cnst22)/1000000)"

"l0=0"
"l10=0"

"l3=td1/4"


"acqt0=0"
baseopt_echo


1 4u ze1
  4u ze2
  d1

2 d11 do:f1 do:f3
  3m
3 3m
4 9m
5 3m
6 9m
7 3m

;n_hcan
```

```
8 d11 do:f2 do:f3
  d60

  50u UNBLKGRAD
  4u pl1:f1 pl3:f3

  (p1 ph1):f1
  p16:gp1
  d16 fq=cnst22(bf ppm):f2

  (p21 ph1):f3
  d4
  (center (p14:sp3 ph1):f2 (p22 ph1):f3 )
  d4
  (p21 ph2):f3

  (p13:sp2 ph3):f2
  d41
  (center (p14:sp5 ph1):f2 (p22 ph1):f3 )
  d41
  (p14:sp23 ph1):f2
  DELTA1
  (p2 ph1):f1
  d51
  (p14:sp5 ph10):f2
  DELTA2
  (p22 ph1):f3
  d31
  (p13:sp8 ph4):f2

  4u
  p16:gp2
  d16 pl16:f3
  20u cpd3:f3

  if "l0 %2 == 0"
     {
     (p1 ph5):f1
     DELTA3
     (p14:sp5 ph1):f2
     DELTA4
     (p14:sp3 ph1):f2
     DELTA5
     (p2 ph6):f1
     DELTA3
     (p14:sp5 ph10):f2
     }
  else
     {
     (p1 ph7):f1
     (p14:sp5 ph1):f2
     DELTA3
     DELTA4
     (p14:sp3 ph1):f2
     DELTA5
     (p2 ph6):f1
     (p14:sp5 ph10):f2
     DELTA3
     }

  DELTA6
  d12 pl28:f2
  4u BLKGRAD

  goscnp1 ph31 cpd2:f2
  lo to 8 times 3


;c_con
```

```
  4u do:f2 do:f3
  50m
  50u UNBLKGRAD

  p16:gp3
  d16 fq=cnst21(bf ppm):f2

  4u pl1:f1 pl16:f3
  4u cpd3:f3

  (p13:sp2 ph1):f2
  d23
  (center (p2 ph1):f1 (p14:sp3 ph1):f2 )
  d23
  (p13:sp8 ph2):f2

  p16:gp4
  d16

  (p1 ph13):f1
  d32
  (p8:sp13 ph15):f2
  d32
  (p2 ph1):f1
  DELTA7
  (p1 ph14):f1

  if "l10 %2 == 0"
     {
     (p13:sp2 ph1):f2
     DELTA8
     (p14:sp7 ph1):f2
     DELTA8
     (center (p2 ph1):f1 (p14:sp3 ph1):f2 )
     DELTA8
     (p14:sp7 ph1):f2
     DELTA8 pl19:f1
     }
  else
     {
     (p13:sp2 ph16):f2
     d22
     (p14:sp7 ph1):f2
     DELTA9
     (center (p2 ph1):f1 (p14:sp3 ph1):f2 )
     DELTA8
     DELTA8 pl19:f1
     (p14:sp7 ph1):f2
     }

  4u BLKGRAD

  go2=2 ph30 cpd1:f1

  d11 do:f1 do:f3 wr2 #1 if2 #1 zd2

;exp_f2 ipap
  3m iu10
  lo to 3 times 2

;exp_f2 phase
  3m dp13
  lo to 4 times 2

  3m wr1 #0 if1 #0 zd1

;exp_f1 ipap
  3m iu0
```

```
;exp_f2 delay
  3m id32
  lo to 5 times 2

;exp_f1 phase
  3m ip3
  lo to 6 times 2

;exp_f1 delay
  3m id31
  3m dd41
  3m id51

  lo to 7 times l3

exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 2 2
ph6=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph7=1 1 3 3
ph10=2

ph13=0 2
ph14=0 0 0 0 2 2 2 2
ph15=0 0 2 2
ph16=3

ph29=0
ph30=0 2 0 2 2 0 2 0
ph31=0 2 2 0 2 0 0 2


;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl16: f3 channel - power level for CPD/BB decoupling
;pl19: f1 channel - power level for CPD/BB decoupling
;pl28: f2 channel - power level for CPD/BB decoupling
;sp2: f2 channel - shaped pulse  90 degree  (on resonance)
;sp3: f2 channel - shaped pulse 180 degree  (on resonance)
;sp5: f2 channel - shaped pulse 180 degree  (C=O off resonance)
;sp7: f2 channel - shaped pulse 180 degree  (Ca off resonance)
;sp8: f2 channel - shaped pulse  90 degree  (on resonance)
;                    for time reversed pulse
;sp13: f2 channel - shaped pulse 180 degree  (adiabatic)
;sp23: f2 channel - shaped pulse 180 degree  (Cali off resonance)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p8 : f2 channel - 180 degree shaped pulse for inversion (adiabatic)
;p13: f2 channel -  90 degree shaped pulse
;p14: f2 channel - 180 degree shaped pulse
;p16: homospoil/gradient pulse                          [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;d1 : relaxation delay as executed; 1-5 * T1
;d3 : 1/(n*J(CH)), n = cnst12
;d4 : 1/(4J(CH))
;d11: delay for disk I/O                                [30 msec]
;d12: delay for power switching                         [20 usec]
;d16: delay for homospoil/gradient recovery
;d22: 1/(4J(COCa))                                      [4.5 msec]
;d23: 1/(4J(NCO))                                       [12.4 msec]
;d25: 1/(4J'(NCO))                                      [16.6 msec]
;d27: 1/(2J(CaCb))                                      [14.3 msec]
;d31 : incremented delay (2D, exp 1) = d27/2-d3-p22/2
```

```
;d32 : incremented delay (2D, exp 2)             [3 usec]
;d41: decremented delay (2D, exp 1) = d27/2-larger(p14,p22)/2
;d51: incremented delay (2D) = d23-d27/2-p14/2-p2/2
;d58: = AQ of exp 1 (15N)
;d59: = AQ of exp 2 (13C)
;d60: relaxation delay for exp 1
;cnst2: = J(CH)                                          [145 Hz]
;cnst12: for multiplicity selection = 4 for all but Gly, 5 including Gly
;cnst21: CO chemical shift (offset, in ppm)
;cnst22: Calpha chemical shift (offset, in ppm)
;cnst23: Caliphatic chemical shift (offset, in ppm)
;cnst26: Call chemical shift (offset, in ppm)          [101 ppm]
;o1p: CO chemical shift (cnst21)
;l0 : flag to switch between inphase and antiphase (exp 1)
;l3 : loop for 2D experiment = td1/4
;l10: flag to switch between inphase and antiphase (exp 2)
;inf1: 1/SW(Ca) = 2 * DW(Ca)
;in2 : 1/SW(N) = 2 * DW(N) (inf1 of exp 2)
;in31: 1/(4 * SW(Ca)) = (1/2) DW(Ca)
;nd31: 4
;in32: 1/(2 * SW(N)) = DW(N)
;nd32: 2
;in42: = in32
;in52: = in32
;ns: 4 * n
;ds: >= 32
;td1: number of experiments in F1
;       (number of experiments after IPAP processing)
;FnMODE: States-TPPI (or TPPI) in F1
;cpd1: decoupling according to sequence defined by cpdprg1
;cpd2: decoupling according to sequence defined by cpdprg2: p5m4sp180
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd1: f1 channel - 90 degree pulse for decoupling sequence
;pcpd2: f2 channel - 180 degree pulse for decoupling sequence
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz1: 50%
;gpz2: 40%
;gpz3: 60%
;gpz4: 30%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100

;use AU-program splitcomb [ipap 2] to process data

;$Id: $
```